# Computation of local and absolute coordinates of the center of instantaneous rotation

Author: Ettore Pennestri' pennestri@mec.uniroma2.it

References: A. Di Benedetto, E. Pennestri', Introduzione alla Cinematica dei Meccanismi, Casa Editrice Ambrosiana, vol. II

```
(%i1) kill(all);
```

(%o0) *done*

## 1 Nomenclature

- P: Center of instantaneous rotation (c.i.r.)
- (xP,yP): Coordinates of point P in the moving reference system o-xy
- (a,b): Absolute coordinates of the origin of the moving reference system o-xy
- theta: angle between the abscissae axes of the two reference systems. The angle is measured c.c.w. from X to x.
- (Px,Py): Absolute coordinates of c.i.r.
- (vPx,vPy): Cartesian velocity components of P
- ds: Infinitesimal arc length on the moving centrode
- dS: Infinitesimal arc length on the fixed centrode
- ids: Complex number whose real and imaginary parts are the cartesian components of ds
- idS: Complex number whose real and imaginary parts are the cartesian components of dS

## 2 Computation of coordinate of P

Define the transform equations between moving and fixed Cartesian reference system

```
(%i1) orig_mov:matrix([a(theta)],[b(theta)]);
      A: matrix(
       [cos(theta),-sin(theta)],
       [sin(theta),cos(theta)]
      );
      P_mov:matrix([xP],[yP]);
      P:orig_mov+A.P_mov;
      Px:P[1][1];
      Py:P[2][1];
```

(%o1) $\begin{pmatrix} \mathrm{a}(\theta) \\ \mathrm{b}(\theta) \end{pmatrix}$

(%o2) $\begin{pmatrix} \cos(\theta) & -\sin(\theta) \\ \sin(\theta) & \cos(\theta) \end{pmatrix}$

(%o3) $\begin{pmatrix} xP \\ yP \end{pmatrix}$

(%o4) $\begin{pmatrix} -\sin(\theta)\; yP + \cos(\theta)\; xP + \mathrm{a}(\theta) \\ \cos(\theta)\; yP + \sin(\theta)\; xP + \mathrm{b}(\theta) \end{pmatrix}$

(%o5) $-\sin(\theta)\; yP + \cos(\theta)\; xP + \mathrm{a}(\theta)$

(%o6) $\cos(\theta)\; yP + \sin(\theta)\; xP + \mathrm{b}(\theta)$

Compute the velocity components of P (c.i.r.)

```
(%i7) vPx:diff(Px,theta);
      vPy:diff(Py,theta);
```

(%o7) $-\cos(\theta)\; yP - \sin(\theta)\; xP + \frac{d}{d\,\theta}\,\mathrm{a}(\theta)$

(%o8) $-\sin(\theta)\; yP + \cos(\theta)\; xP + \frac{d}{d\,\theta}\,\mathrm{b}(\theta)$

Set equal to zero these components and solve the system of equations w.r.t. xP and yP

```
(%i9) sols:linsolve([vPx, vPy], [xP,yP]);
```

(%o9) $[xP = \frac{\sin(\theta)\;\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right) - \cos(\theta)\;\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)}{\sin(\theta)^2 + \cos(\theta)^2}, yP = \frac{\sin(\theta)\;\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right) + \cos(\theta)\;\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)}{\sin(\theta)^2 + \cos(\theta)^2}]$

Coordinates of c.i.r. in the moving system

```
(%i10) xp1:trigsimp(rhs(sols[1]))$
       yp1:trigsimp(rhs(sols[2]))$
       print("Coordinates of P in the moving system")$
       print("xp=",xp1)$
       print("yp=",xp1)$
```

Coordinates of P in the moving system

$$xp = \sin(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right) - \cos(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)$$

$$yp = \sin(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right) - \cos(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)$$

Coordinates of c.i.r. in the absolute reference system

```
(%i15) Px:trigsimp(subst(xp1,xP,Px))$
       Px:trigsimp(subst(yp1,yP,Px))$
       Py:trigsimp(subst(xp1,xP,Py))$
       Py:trigsimp(subst(yp1,yP,Py))$
       print("Coordinates of P in the fixed system")$
       print("Px=",Px)$
       print("Py=",Py)$
```

Coordinates of P in the fixed system

$$Px = \mathrm{a}(\theta) - \frac{d}{d\,\theta}\,\mathrm{b}(\theta)$$

$$Py = \frac{d}{d\,\theta}\,\mathrm{a}(\theta) + \mathrm{b}(\theta)$$

## 3 Centrodes

- Fixed centrode: Locus of points on the fixed system occupied by P
- Moving centrode: Locus of points on the moving plane that will be c.i.r.

One can demonstrate that:

- since the moving centrode rolls on the fixed centrode the lengths of their arcs between corresponding points are equal;
- the two curves have a common tangent in P.

Computation of infinitesimal arc length components on the moving centrode

```
(%i22) dsx:collectterms(diff(xp1,theta),cos(theta),sin(theta));
       dsy:collectterms(diff(yp1,theta),cos(theta),sin(theta));
       ds:trigsimp(sqrt(dsx^2+dsy^2));
```

(%o22) $\cos(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)-\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)\right)+\sin(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)+\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)$

(%o23) $\sin(\theta)\left(\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)-\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)+\cos(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)+\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)$

(%o24) $\sqrt{\left(\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)\right)^2-2\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)\left(\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)\right)+\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)^2+2\left(\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)+\left(\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)^2+\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)^2}$

Computation of infinitesimal arc length components on the fixed centrode

```
(%i25) dSX:diff(Px,theta);
       dSY:diff(Py,theta);
       dS:trigsimp(sqrt(dSX^2+dSY^2));
```

(%o25) $\frac{d}{d\,\theta}\,\mathrm{a}(\theta)-\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)$

(%o26) $\frac{d}{d\,\theta}\,\mathrm{b}(\theta)+\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)$

(%o27) $\sqrt{\left(\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)\right)^2-2\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)\left(\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)\right)+\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)^2+2\left(\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)+\left(\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)^2+\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)^2}$

Check that ds (infinitesimal arc length on the moving centrode) and dS (infinitesimal arc length on the fixed centrode) are equal.

```
(%i28) is(equal (ds,dS));
```

(%o28) *true*

```
(%i29) ids:dsx+%i*dsy;
       idS:dSX+%i*dSY;
       ids:expand(exponentialize(ids));
       idS:expand(exponentialize(idS));
```

(%o29) $i\left(\sin(\theta)\left(\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)-\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)+\cos(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)+\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)\right)+$
$\cos(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)-\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)\right)+\sin(\theta)\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)+\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)$

(%o30) $-\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)+i\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)+\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)+\frac{d}{d\,\theta}\,\mathrm{a}(\theta)$

(%o31) $-e^{-i\,\theta}\left(\frac{d^2}{d\,\theta^2}\,\mathrm{b}(\theta)\right)+i\,e^{-i\,\theta}\left(\frac{d}{d\,\theta}\,\mathrm{b}(\theta)\right)+i\,e^{-i\,\theta}\left(\frac{d^2}{d\,\theta^2}\,\mathrm{a}(\theta)\right)+e^{-i\,\theta}\left(\frac{d}{d\,\theta}\,\mathrm{a}(\theta)\right)$

(%o32) $-\frac{d^2}{d\,\theta^2}\,\mathrm{b}\left(\theta\right)+i\,\left(\frac{d}{d\,\theta}\,\mathrm{b}\left(\theta\right)\right)+i\,\left(\frac{d^2}{d\,\theta^2}\,\mathrm{a}\left(\theta\right)\right)+\frac{d}{d\,\theta}\,\mathrm{a}\left(\theta\right)$

```
(%i33) tt:ratsimp(ids/idS);
```

(%o33) $e^{-i\,\theta}$

We observe that apparently the tangents to the centrodes form an angle of -theta. However, one should consider that ds is expressed in a reference system which is rotated of the angle theta. We conclude that the polodes share the same tangent at c.i.r.